

# Métodos Numéricos Aplicados em Camada Limite Bi-dimensional

## Índice

Métodos Numéricos Aplicado a Camada Limite Bi-Dimensional .....	2
Modelo de Diferenças Finitas .....	3
Esquemas Implícitos x Explícitos.....	4
Esquema Explícito .....	5
Esquema Implícito I.....	8
Esquema Implícito II.....	9
Implementação Numérica do Esquema Implícito I .....	11
Referências & Leitura Complementar .....	24

Prof. Eugênio Spanó Rosa  
FEM-DE UNICAMP  
erosa@fem.unicamp.br

## Métodos Numéricos Aplicado a Camada Limite Bi-Dimensional

Os métodos integrais discutidos na seção anterior são simples de usar porém com uma precisão limitada ( $\pm 15\%$ ). Por sua vez, as soluções similares são precisas entretanto possuem aplicações restritas. Se o problema a ser abordado é de natureza não-similar e for requerido boa precisão nos resultados então ele se torna um caso típico para desenvolvimento em computadores digitais dos modelos numéricos.

Por conveniência, as equações para a camada limite laminar bi-dimensional em regime permanente são repetidas:

$$\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} = 0$$

$$u \frac{\partial u}{\partial x} + v \frac{\partial u}{\partial y} = U \frac{dU}{dx} + \nu \frac{\partial^2 u}{\partial y^2}$$

Estas equações são de natureza parabólica na direção  $x$  de maneira que todos os modelos numéricos resolvem o campo de velocidades  $u$  na medida que marcham ao longo do eixo  $x$ . As condições iniciais e de contorno para os métodos são: (a) perfil de velocidades conhecido a montante,  $u(0,y)$  e  $v(0,y)$ ; (b) conhecida a velocidade do escoamento externo  $U(x)$ ; (c) conhecida as condições na parede  $u(x,0) = 0$  e  $v(x,0) = v_w$ .

Referente aos tipos de modelos numéricos, os parabólicos ou "modelos de marcha", são os modelos mais simples de se implementar e conseqüentemente existe uma vasta literatura sobre o assunto com diferentes proposições de implementação. Os métodos mais populares são três: a) método Keller-Box implementado por Cebeci and Bradshaw (1977); b) o método de Crank-Nicholson implementado para problemas em camada limite por Blottner (1970); e, c) o método dos volumes finitos desenvolvido por Patankar e Spalding (1970). Uma excelente revisão destes métodos pode ser encontrada no Numerical Heat Transfer Handbook, Minkowycz et alli (1988). Estes códigos, por questões de eficiência numérica tendem a usar as variáveis primitivas de forma transformada, isto é, utilizando formulação de função corrente e/ou procedendo as transformações de Falkner-Skan. Uma das vantagens explícitas deste procedimento é que a equação da massa não é considerada porque ela é automaticamente satisfeita entretanto, estas variáveis transformadas e suas discretizações tendem a deixar os passos intermediários dos programas um tanto quanto obscuro relativo as variações das grandezas.

Para os propósitos do presente texto será apresentado um método simples utilizando diferenças finitas baseado diretamente nas variáveis primitivas,  $u(x,y)$  e  $v(x,y)$ . Através de sua modelagem e implementação numérica pretende-se lançar bases para se avançar para métodos mais sofisticados.

## Modelo de Diferenças Finitas

O modelo a ser apresentado utiliza uma grade retangular como mostrado na Fig. 1. Os sub-índices de  $u_{i,j}$  referem-se a velocidade  $u$  no nó  $(i,j)$  que está localizado na coordenada  $(x,y) = [i.\Delta x, j.\Delta y]$ . Então  $(i,j) = (0,0)$  refere-se a origem do sistema coordenado,  $(x,y) = (0,0)$ . Os passos na direção  $x$  e  $y$ ,  $\Delta x$ , e  $\Delta y$ , não necessariamente devem ser iguais. A velocidade  $v_{i,j}$  ou outras propriedades como temperatura, estão localizadas nos mesmos pontos da grade. Todos as grandezas na estação  $i-1$  são tidas como conhecidas, "valores prévios". O método utiliza estes valores prévios para proceder o cálculo dos "proximos valores"; este procedimento de cálculo caracteriza o processo de marcha ao longo da direção  $x$ . O tamanho da malha é definido pelas variáveis  $Limx$  e  $Limy$  onde  $i = Limx$  é a saída do domínio,  $j=Limy$  é o escoamento externo,  $i = 0$  é a entrada no domínio e  $j = 0$  é a parede sólida.




Fig. 1 - Grade de diferenças finitas para camada limite bi-dimensional.

A discretização das equações do momento e massa podem, de maneira genérica, serem colocadas como segue:

$$\begin{aligned}
& \left[ T \cdot u_{i-1,j} + (1-T) \cdot u_{i,j} \right] \cdot \frac{u_{i,j} - u_{i-1,j}}{\Delta x} \\
& + \left[ T \cdot v_{i-1,j} + (1-T) \cdot v_{i,j} \right] \cdot \left\{ R \cdot \frac{u_{i-1,j+1} - u_{i-1,j-1}}{2\Delta y} + (1-R) \frac{u_{i,j+1} - u_{i,j-1}}{2\Delta y} \right\} \\
& = \\
& \frac{1}{2} \frac{U_i^2 - U_{i-1}^2}{\Delta x} \\
& + v \cdot \left\{ S \cdot \frac{u_{i-1,j-1} - 2 \cdot u_{i-1,j} + u_{i-1,j+1}}{\Delta y^2} + (1-S) \cdot \frac{u_{i,j-1} - 2 \cdot u_{i,j} + u_{i,j+1}}{\Delta y^2} \right\}
\end{aligned} \tag{1}$$

$$\left[ M \cdot \frac{u_{i,j} - u_{i-1,j}}{\Delta x} + (1-M) \cdot \frac{u_{i,j+1} - u_{i-1,j+1}}{\Delta x} \right] + \left[ N \cdot \frac{v_{i,j} - v_{i,j-1}}{\Delta y} + (1-N) \cdot \frac{v_{i-1,j} - v_{i-1,j-1}}{\Delta y} \right] = 0 \tag{2}$$

onde T, S, R, M e N são constantes que variam entre 0 e 1 inclusive, por exemplo  $0 \leq S \leq 1$ . Os valores que estas constantes podem assumir afetam, diretamente, a ordem de aproximação das diferenças finitas e a estabilidade do esquema. Na seção que segue são discutidos estes tópicos.

### Esquemas Implícitos x Explícitos

Nesta seção são discutidos dois esquemas resultantes das equações discretizadas, Eqs. (1) e (2). Dependendo da escolha da constantes S a equação discretizada se torna implícita ou explícita. Basicamente a diferença entre estes esquemas é com relação a estabilidade numérica e ao esforço computacional para resolve-los. Enquanto que o esquema explícito a solução é obtida de maneira direta através de uma equação algébrica ele é limitado quanto aos passos nas direções x e y que o método pode avançar por restrições de estabilidade. Já por outro lado, o esquema implícito é intrinsecamente estável independentemente do tamanho dos passos nas direções x e y porém, é necessário resolver um sistema de equações algébricas simultaneamente.

### Esquema Explícito

Quando as constantes T, R e S são iguais a unidade, a Eq. (1) se reduz a:

$$\begin{aligned} & \left[ u_{i-1,j} \right] \cdot \frac{u_{i,j} - u_{i-1,j}}{\Delta x} + \left[ v_{i-1,j} \right] \cdot \left\{ \frac{u_{i-1,j+1} - u_{i-1,j-1}}{2\Delta y} \right\} \\ & = \\ & \frac{1}{2} \frac{U_i^2 - U_{i-1}^2}{\Delta x} + v \cdot \left\{ \frac{u_{i-1,j-1} - 2 \cdot u_{i-1,j} + u_{i-1,j+1}}{\Delta y^2} \right\} \end{aligned} \quad (3)$$

Analisando-se os termos da Eq. (3) nota-se que o primeiro termo usa uma "diferença adiantada" para modelar  $\partial u / \partial x$  com um erro de truncamento da ordem de  $\Delta x$ . O segundo termo utiliza uma diferença central para representar  $\partial u / \partial y$  com um erro de truncamento da ordem de  $\Delta y^2$ . O terceiro termo é o gradiente de pressão usando a forma  $UdU/dx = d(U^2/2)/dx$  e é escrito como uma diferença adiantada; note que  $U_i$  é conhecido porque  $U(x)$  é a velocidade externa a camada limite. Finalmente, o quarto termo é uma diferença central de  $\partial^2 u / \partial y^2$  com um erro de truncamento da ordem de  $\Delta y^2$ . Note que todos os termos são aproximações das derivadas sobre o nó  $(i-1)$ , o que é uma boa prática para se obter precisão numérica. Na expressão acima a única incógnita é a velocidade  $u_{ij}$  que pode ser determinada por:

$$u_{i,j} = (\alpha - \beta) \cdot u_{i-1,j+1} + (1 - 2\alpha) \cdot u_{i-1,j} + (\alpha + \beta) \cdot u_{i-1,j-1} + \frac{U_i^2 - U_{i-1}^2}{2u_{i-1,j}} \quad (4)$$

onde

$$\alpha = \frac{v\Delta x}{u_{i-1,j}\Delta y^2} \quad \text{e} \quad \beta = \frac{v_{i-1,j}\Delta x}{2u_{i-1,j}\Delta y} \quad (5)$$

Os termos do lado direito da Eq. (4) são todos conhecidos porque são valores previamente calculados. Portanto pode-se resolver para  $u_{ij}$  diretamente e devido a este fato o modelo é denominado *explícito*. Entretanto a relação algébrica de recorrência dada pela Eq. (4) requer, para estabilidade numérica, que todos os coeficientes dos valores prévios de  $u_{i-1}$  sejam positivos. Isto implica  $\alpha < 1/2$  e  $\beta < \alpha$ . Estas restrições limitam os passos nas direções x e y da malha:

$$\Delta x < \frac{u_{\min} \Delta y^2}{2v} \quad \Delta y < \frac{2v}{|v_{\max}|} \quad (6)$$

Na Eq. (6) é necessário tomar o valor absoluto de  $v$  porque ele pode ser positivo ou negativo dependendo, respectivamente, se o gradiente de pressão for adverso ou favorável.

Usualmente os valores mínimos de  $v$  ocorrem próximos da parede e os máximos na camada externa. Durante os cálculos deste modelo explícito é necessário monitorar os valores de  $v$  para ver se a Eq. (6) está sendo satisfeita.

Com os valores de  $u$  determinados para a estação  $i$  pode-se então determinar os valores da velocidade  $v$  nesta estação. Isto é realizado através da equação da conservação da massa. A forma mais simples, porém não conveniente do ponto de vista de precisão numérica, é dada fazendo  $N = M = 1$ :

$$\left[ \frac{u_{i,j} - u_{i-1,j}}{\Delta x} \right] + \left[ \frac{v_{i,j} - v_{i,j-1}}{\Delta y} \right] = 0 \quad (7)$$

Começando com  $j = 1$  para calcular  $v_{i,j}$  porque  $v_{i,0} = v_w$  é conhecido. Entretanto, observando Eq. (7) nota-se que a aproximação para  $\partial u / \partial x$  está no nível  $j$  enquanto que para  $\partial v / \partial y$  no nível  $j-1/2$ , como sugere a Fig. 2. Este tipo de esquema gera pouca precisão numérica.




Fig. 2 - Representação geométrica na malha de diferenças finitas da Eq. (7)

Um remédio para isto é fazer a constante  $M = 1/2$  enquanto que  $N = 1$ , a equação da massa passa então a ser:

$$\frac{1}{2} \left[ \frac{u_{i,j} - u_{i-1,j}}{\Delta x} + \frac{u_{i,j+1} - u_{i-1,j+1}}{\Delta x} \right] + \left[ \frac{v_{i,j} - v_{i,j-1}}{\Delta y} \right] = 0 \quad (8)$$

Neste caso as derivadas são calculadas no nível  $j-1/2$  resultando em uma melhora da precisão numérica em relação ao esquema proposto pela Eq. (7), veja Fig. 3. Como a Eq. (8) requer um esforço computacional relativamente pequeno comparado com a Eq. (7), este esquema para determinação da velocidade  $v$  é recomendado.




Fig. 3 - Representação geométrica na malha de diferenças finitas da Eq. (8)

Uma última observação com relação ao sistema de equações discretizadas é com relação a condição inicial para dar início ao processo de marcha. Na fronteira de entrada do domínio,  $x = 0$ , é necessário especificar o perfil de velocidade  $u$  de maneira precisa entretanto a especificação do perfil de  $v$  já não é. Para especificação do perfil  $v$  pode-se adotar o resultante de uma solução das equações de Falkner-Skan se o caso em estudo for de escoamento em cunhas ou com estagnação; ou funções arbitrárias e mesmo zeros. A razão para isto é que o processo numérico atualiza os valores de  $v$  em cada estação  $i$  a partir da equação da massa.

### Esquema Implícito I

Fazendo as constantes  $T = R = 1$  e  $S = 0$  a Eq. (1) fica sendo:

$$\begin{aligned} & \left[ u_{i-1,j} \right] \cdot \left\{ \frac{u_{i,j} - u_{i-1,j}}{\Delta x} \right\} + \left[ v_{i-1,j} \right] \cdot \left\{ \frac{u_{i-1,j+1} - u_{i-1,j-1}}{2\Delta y} \right\} \\ & = \end{aligned} \quad (9)$$

$$\frac{1}{2} \frac{U_i^2 - U_{i-1}^2}{\Delta x} + v \cdot \left\{ \frac{u_{i,j-1} - 2 \cdot u_{i,j} + u_{i,j+1}}{\Delta y^2} \right\}$$

Na expressão acima as incógnitas são os "próximos valores" das velocidades  $u_{i,j-1}$ ,  $u_{ij}$  e  $u_{i,j+1}$ . A solução da velocidade  $u$  para a estação  $i$  só é possível através da solução simultânea das equações algébricas lineares definidas pela Eq. (9) varrendo todos os nós da estação  $i$  inclusive as condições de contorno. O sistema de equações resultante é obtido através da Eq. (10),  $\alpha$  e  $\beta$  estão definidos na Eq. (5); destaca-se a natureza tri-diagonal definida pelo lado esquerdo da equação com coeficientes  $-\alpha$ ,  $1+2\alpha$ ,  $-\alpha$ , já o lado direito são os valores conhecidos da seção prévia.

$$-\alpha \cdot u_{i,j+1} + (1 + 2\alpha) \cdot u_{i,j} - \alpha \cdot u_{i,j-1} = + \frac{U_i^2 - U_{i-1}^2}{2u_{i-1,j}} + u_{i-1,j} - \beta(u_{i-1,j+1} - u_{i-1,j-1}) \quad (10)$$

Devido ao fato que a solução de  $u$  na estação  $i$  é obtida através da solução do sistema linear de equações algébricas, este esquema discretizante é denominado **implícito**. O benefício que ele resulta está no fato dele ser incondicionalmente estável e portanto o tamanho dos passos  $\Delta x$  e  $\Delta y$  são independentes. Destaca-se que é o valor que a constante  $S$  assume o responsável para transformar o esquema implícito; valores de  $S$  dentro do intervalo:  $0 \leq S \leq 1/2$  são suficientes para tornar a discretização incondicionalmente estável.

Apesar da restrição do tamanho dos passos estar relaxada neste esquema, deve tomar uma precaução quanto a sua escolha tendo em mente que as equações discretizadas refletem uma aproximação numérica das derivadas. Neste sentido recomenda-se  $\Delta y$  ser de um valor tal que pelo menos 20 pontos fiquem dentro da região da camada limite. De maneira semelhante, recomenda-se que  $\Delta x$  seja tal que as variações de  $u$  entre uma estação  $i-1$  e  $i$  sejam de até 5%.

Uma vez resolvido as velocidades  $u$  na estação  $i$ , as velocidades  $v$  são obtidas diretamente através da Eq. (8) que, re-escrita de forma mais conveniente fica sendo:

$$v_{i,j} = \frac{\Delta y}{2\Delta x} \left[ (u_{i,j} - u_{i-1,j}) + (u_{i,j+1} - u_{i-1,j+1}) \right] + v_{i,j-1} \quad (11)$$

A denominação de esquema implícito I se deve ao fato que as aproximações numéricas das derivadas associadas à Eq. (10) refere-se ao nível  $j$  entretanto, se for observado o nível  $i$  constata-se que o termo  $\partial u / \partial y$  é avaliado no nível  $i-1$  enquanto que o termo  $\partial^2 u / \partial y^2$  é avaliado no nível  $i$ . Este tipo de abordagem faz com que haja uma perda da precisão do método devido ao fato que as derivadas são obtidas por aproximações de



diferentes níveis. De forma similar o mesmo problema ocorre com a equação da massa, Eq. (8), já discutido anteriormente. Uma maneira de se melhorar esta deficiência é apresentada no esquema implícito II

### Esquema Implícito II

Fazendo as constantes:  $T = R = S = 1/2$ , a Eq. (1) fica sendo:

$$\begin{aligned}
 & \left[ \frac{1}{2} (u_{i-1,j} + u_{i,j}) \right] \cdot \left\{ \frac{u_{i,j} - u_{i-1,j}}{\Delta x} \right\} \\
 & + \left[ \frac{1}{2} (v_{i-1,j} + v_{i,j}) \right] \cdot \left\{ \frac{1}{2} \left[ \frac{u_{i-1,j+1} - u_{i-1,j-1}}{2\Delta y} + \frac{u_{i,j+1} - u_{i,j-1}}{2\Delta y} \right] \right\} \\
 & = \\
 & \frac{1}{2} \frac{U_i^2 - U_{i-1}^2}{\Delta x} \\
 & + v \cdot \left\{ \frac{1}{2} \left[ \frac{u_{i-1,j-1} - 2 \cdot u_{i-1,j} + u_{i-1,j+1}}{\Delta y^2} + \frac{u_{i,j-1} - 2 \cdot u_{i,j} + u_{i,j+1}}{\Delta y^2} \right] \right\}
 \end{aligned}
 \tag{12}$$

Note que neste esquema as derivadas e as velocidades  $u$  e  $v$  são estimadas para o nó  $i-1/2, j$ , conforme sugerido pela Fig. 4. Este esquema é mais favorável à precisão numérica que o esquema I, Eq. (9), porque todas as estimativas numéricas são realizadas no mesmo nível. Assim como o caso anterior ele é implícito com erro de truncamento nas direções  $x$  e  $y$  proporcionais a  $\Delta x$  e  $\Delta y^2$ . Ele foi originalmente proposto por Blottner, F.G. (1970).




Fig. 4 - Representação geométrica na malha de diferenças finitas da Eq. (12)

A melhoria da precisão numérica na discretização da Eq. (12) ocorre as custas de um esforço computacional maior em relação ao esquema proposto pela Eq. (9). Observando-se a Eq. (12) nota-se que seu primeiro termo introduz uma não linearidade no sistema algébrico de equações pois introduz o produto de uma variável ainda não conhecida,  $u_{ij} \cdot u_{ij}$ . Para resolver este sistema como um sistema de equações algébricas lineares deve-se recorrer a uma solução iterativa para cada estação  $i$ . Isto é, para cada estação  $i$  o processo de cálculo incia-se com os valores "antigos" da velocidade  $u$  para obter uma nova velocidade  $u$  que será utilizada para a próxima iteração na mesma estação  $i$  até que a diferença entre o valor novo e o valor antigo fiquem dentro de uma tolerância especificada.

Especificando um contador de iterações na estação  $i$  pelo índice  $p$ , a Eq. (12) é transformada em uma equação linear substituindo-se os valores de  $u_{i,j}$  e  $v_{i,j}$  por aqueles disponíveis na iteração  $p-1$ . Denotando as quantidades a serem calculadas no ciclo  $p$  das iterações pelo super-escrito  $p$ , a Eq. (12) pode ser re-escrita na forma:

$$-A_{i-1,j} \cdot u_{i,j+1}^p + B_{i-1,j} \cdot u_{i,j}^p - C_{i-1,j} \cdot u_{i,j-1}^p = D_{i-1,j} \quad (13)$$

onde os coeficientes  $A_{i-1,j}$ ,  $B_{i-1,j}$  e  $C_{i-1,j}$  dependem dos valores já disponíveis na estação  $i-1$  e dos valores calculados na iteração  $(p-1)$  na estação  $i$ . Desta forma o conjunto de equações formado, quando Eq. (13) é aplicada para todos os valores de  $j$  na estação  $i$ , resulta em um sistema linear na forma tri-diagonal.

De forma similar, a equação da massa pode ter sua precisão numérica melhorada em relação ao esquema proposto pela Eq. (8) fazendo-se as constantes  $M = N = 1/2$  na Eq. (2) que passa a ser:

$$\left[ \frac{1}{2} \left( \frac{u_{i,j} - u_{i-1,j}}{\Delta x} + \frac{u_{i,j+1} - u_{i-1,j+1}}{\Delta x} \right) \right] + \left[ \frac{1}{2} \left( \frac{v_{i,j} - v_{i,j-1}}{\Delta y} + \frac{v_{i-1,j} - v_{i-1,j-1}}{\Delta y} \right) \right] = 0 \quad (13)$$

Neste esquema as aproximações para as derivadas  $\partial u / \partial x$  e  $\partial v / \partial y$  passam a ser simultaneamente estimadas no nível  $i-1/2, j-1/2$ , conforme sugere a Fig. 5.




Fig. 5 - Representação geométrica na malha de diferenças finitas da Eq. (13)

A velocidade  $v_{ij}$  é então obtida de maneira explícita re-escrevendo Eq. (13) na forma:

$$v_{i,j} = v_{i,j-1} - (v_{i-1,j} - v_{i-1,j-1}) - \frac{\Delta y}{\Delta x} \cdot [(u_{i,j} - u_{i-1,j}) + (u_{i,j+1} - u_{i-1,j+1})] \quad (14)$$

### Implementação Numérica do Esquema Implícito I

Nesta seção é implementado numericamente a rotina proposta como esquema implícito I para solução da camada limite não similar que ocorre no escoamento ao redor de um cilindro. Esta seção foi extraída na íntegra do artigo "Metodo de Diferenças Finitas Aplicado a Camada Limite Usando Mathematica", Rosa, E.S., Scalon, V. e Mazza, R.A (1995)

## Método de Diferenças Finitas

## Aplicado a Camada Limite Usando Mathematica®

### INTRODUÇÃO

Tem surgido, nos últimos anos, uma nova geração de softwares que influenciará marcadamente os meios atuais de ensino e pesquisa. Um destes pacotes é o Mathematica® que permite a criação de algoritmos através de manipulações numéricas e simbólicas de uma maneira unificada, assim como a geração de gráficos. Informações detalhadas encontram-se Wolfram (1991).

O presente trabalho pretende desenvolver uma aplicação no ambiente do Mathematica® para solução numérica de problemas de camada limite hidrodinâmica. O enfoque é dado ao aspecto educacional do pacote e aplicado na solução destes problemas. Com as facilidades para implementação e solução de esquemas discretizantes pode-se explorar a natureza física de problemas não similares em camada limite, estabelecer comparações com soluções analíticas e métodos integrais, estudar controle de camadas limites com sucção e injeção de massa e desenvolver projetos de cursos com escopo mais abrangentes, isto é, não restritos somente as soluções analíticas e aos métodos integrais.

### EQUAÇÕES DA CAMADA LIMITE

O estudo é restrito aos problemas bidimensionais da camada limite hidrodinâmica de escoamentos incompressíveis e laminares. A solução é implementada para o escoamento ao redor de um cilindro de raio  $R$ , por se tratar de um caso de grande aplicação técnica e que, conhecidamente, não possui uma solução similar. A Fig.1 ilustra o sistema de coordenadas utilizado;  $x$  e  $y$  são, respectivamente, as coordenadas paralelas e transversais a direção do escoamento;  $u$  e  $v$  são as velocidades nas direções  $x$  e  $y$ . Destaca-se que a origem das coordenadas é coincidente com o ponto de estagnação frontal do cilindro.




Fig. 1 - Representação esquemática do sistema de coordenadas no cilindro.

Procedendo a adimensionalização das variáveis através das Eqs. (1) e (2)

$$x^* = \frac{x}{R}, \quad y^* = \frac{y}{R} \sqrt{\text{Re}_\infty} \quad (1)$$

$$u^* = \frac{u}{U_\infty}, \quad v^* = \frac{v}{U_\infty} \sqrt{\text{Re}_\infty}, \quad U^* = \frac{U}{U_\infty}, \quad P^* = \frac{P}{\rho_\infty U_\infty^2} \quad (2)$$

chega-se a forma adimensional das equações do momento e massa (White 1991):

$$u^* \frac{\partial u^*}{\partial x^*} + v^* \frac{\partial u^*}{\partial y^*} = \frac{1}{2} \frac{dU^{*2}}{dx^*} + \frac{\partial^2 u^*}{\partial y^{*2}} \quad (3)$$

$$\frac{\partial u^*}{\partial x^*} + \frac{\partial v^*}{\partial y^*} = 0 \quad (4)$$

onde o superescrito (\*) refere-se a variável adimensional, U a velocidade externa a camada limite,  $U_\infty$  a velocidade de aproximação ao cilindro e  $\text{Re}_\infty$  o número de Reynolds definido por  $\text{Re}_\infty = U_\infty \cdot R / \nu$ . As Eqs. (3) e (4) satisfazem as condições de contorno:

$$\begin{aligned} u^*(x^*, 0) &= 0 && \text{(não deslizamento)} \\ v^*(x^*, 0) &= 0 && \text{(sem sucção ou injeção massa)} \\ u^*(x^*, \infty) &= U^* && \text{(casamento com escoamento externo)} \end{aligned}$$

onde a velocidade externa,  $U^*$ , é dada através do ajuste experimental obtido por Hiemenz para um cilindro com  $\text{Re}_\infty = 9500$  (White 1991):

$$U^* = 1.814 \cdot x^* - 0.271 \cdot x^{*3} - 0.0471 \cdot x^{*5} \quad (5)$$

As condições iniciais, ou condições de partida, são aquelas para o escoamento em um ponto de estagnação bidimensional. A obtenção dos perfis u e v é realizada através da solução por similaridade das equações de Falkner-Skan quando o parâmetro  $m = 1$  (White 1991),

$$F''' + F \cdot F' + \frac{2m}{m+1} \cdot (1 - F^2) = 0 \quad (6)$$

sujeita as condições de contorno:

$$F(0) = 0; \quad F'(0) = 0 \quad \text{e} \quad F''(0) = 1.23259 \quad (7)$$

Os perfis de velocidade são obtidos através da solução de F e de suas derivadas como segue:

$$u(0, y) = U(0) \cdot F'(\eta) \equiv 0$$

$$v(0, y) = -U(0) \cdot \sqrt{\frac{m+1}{\text{Re}_x}} \left( F(\eta) + \frac{m-1}{m+1} \eta F'(\eta) \right)$$

onde  $\eta$  é a variável de similaridade,

$$\eta = \frac{y}{x} \sqrt{\frac{(m+1) \cdot \text{Re}_x}{2}}$$

Em termos das variáveis adimensionais do problema, os perfis  $u$  e  $v$  se transformam em:

$$u^*(0, y^*) = 0 \quad (8)$$

$$v^*(0, y^*) = -\sqrt{1.814} \cdot F\left[y^* \sqrt{1.814}\right] \quad (9)$$

### ESQUEMA DE DISCRETIZAÇÃO

Existem na literatura uma variedade de esquemas discretizantes para o sistema parabólico de equações em  $x$ , definido pelas Eqs. (5) e (6). Talvez os três esquemas mais populares são: o método de "box" desenvolvido por Keller e implementado por Cebeci e Bradshaw (1977), o método de Crank-Nicolson implementado para solução em camada limite por Blotner (1970) e o método de volumes finitos desenvolvido por Patankar e Spalding (1970). Estes trabalhos utilizam, para melhorar a eficiência dos cálculos, transformações de variáveis utilizando a função corrente, similaridade local com as variáveis de Falkner-Skan, etc. Em geral estes tipos de transformações tendem a obscurecer o que ocorre durante o desenvolvimento do algoritmo.

O objetivo deste trabalho não é a comparação entre os métodos existentes ou o desenvolvimento de um novo método numérico, mas sim, enfatizar o caráter educacional da implementação destas técnicas. Para tanto propõe-se um esquema discretizante utilizando as formas mais simples de diferenças finitas com as variáveis primitivas  $x$ ,  $y$  e  $u$ ,  $v$ , conforme sugerido por White (1991).

A malha adotada é uniformemente espaçada nas direções  $x$  e  $y$  com variações  $\Delta x$  e  $\Delta y$ , respectivamente. As colunas e linhas são denotadas pelos índices  $i$  e  $j$  sendo que a origem está em  $(i, j) = (0, 0)$ . O domínio computacional é mostrado na Fig. 2.

O esquema discretizante utilizado é implícito na velocidade  $u$  com uma aproximação de segunda ordem para  $y$  e de primeira ordem para  $x$ . A Eq. (3) discretizada é escrita por:




Fig.2 - Malha computacional para um esquema de diferenças finitas aplicado a camada limite bidimensional.

$$u_{i-1,j}^* \frac{u_{i,j}^* - u_{i-1,j}^*}{\Delta x^*} + v_{i-1,j}^* \frac{u_{i-1,j+1}^* - u_{i-1,j-1}^*}{2\Delta y^*} = \frac{1}{2} \frac{U_i^{*2} - U_{i-1}^{*2}}{\Delta x^*} + \frac{u_{i,j-1}^* - 2 \cdot u_{i,j}^* + u_{i,j+1}^*}{\Delta y^{*2}} \quad (10)$$

A velocidade  $v$  para o nó  $(i,j)$  é obtida através da Eq. (4) usando uma média do valor discreto de  $\partial u / \partial x$ :

$$v_{i,j}^* = v_{i,j-1}^* - \frac{\Delta y^*}{2 \cdot \Delta x^*} (u_{i,j}^* - u_{i-1,j}^* + u_{i,j-1}^* - u_{i-1,j-1}^*) \quad (11)$$

O campo de velocidades  $u$  no nível  $(i)$  é obtido através da solução simultânea do sistema de equações definido pela Eq. (10). Assim é estabelecido um procedimento de marcha ao longo da direção  $x$  resolvendo-se, coluna a coluna, o domínio computacional conforme sugerido na Fig. 2. Uma vez determinado o campo de velocidades  $u$  no nível  $(i)$ , o campo de velocidades  $v$  é calculado através da Eq. (11). Na malha computacional, Fig. 2, os nós:  $(i, \text{lim}y)$  referem-se às condições do escoamento externo,  $(0, j)$  às condições iniciais e,  $(i, 0)$  às condições de contorno de fronteira sólida. Maiores detalhes sobre o esquema discretizante podem ser encontrados em White (1991).

### IMPLEMENTAÇÃO NUMÉRICA NO MATHEMATICA®

O objetivo desta seção é mostrar a forma de implementação do algoritmo no ambiente do Mathematica®.

É recomendável que, na primeira linha de programação se limpe da memória todas as variáveis que serão utilizadas e, no caso de se utilizar alguma matriz para armazenamento de dados, defini-la como nula. Isto é feito nas linhas de programação:

```
In[1]:=
Remove["Global`*"];
espmom={};
espdes={};
cf      ={};
```

Como no Mathematica® é possível realizar cálculo simbólico, está sendo desenvolvido um procedimento que permite obter equações discretizadas a partir de sua forma diferencial. Ele é baseado em funções de forma de elementos finitos que permite a derivação automática dos coeficientes da diferença finita correspondentes a uma especificada derivada parcial (Mikhailov 1994). A obtenção das Eqs. (10) e (11) é feita utilizando o pacote:

### ■ Discretizações das Eqs. a Serem Utilizadas

```
In[5]:=
<<Unicamp`Discret` (*Carrega o
Pacote*)
```

### ■ Eq. da Conservação de Quantidade de Movimento

```
In[6]:=
eqMovi = Primitiva[u,x,-1]*
DerivDiscrPrim[u,x,dx,1]      +
Primitiva[v,x,-1]*
DerivDiscrPrim[u,y,dy,2]      +
gradPress[x,dx] -
DerivDiscrSeg[u,y,dy] == 0
```

Out[6]:=

$$\text{grad Press}[x, dx] + \frac{u[-1+i, j](-u[-1+i, j] + u[i, j])}{dx} - \frac{u[i, -1+j] - 2u[i, j] + u[i, 1+j]}{dy^2} + \frac{(-u[-1+i, -1+j] + u[-1+i, j+1])v[-1+i, j]}{2dy} == 0$$

### ■ Eq. da Conservação da Massa

```
In[7]:=
eqMassa[j_] = v[j-1] - (0.5*dy)*
(DerivDiscrPrim[u,x,dx,1] +
(DerivDiscrPrim[u,x,dx,1] /. j -
> j - 1))
```



$$\begin{aligned} \text{Out}[6]:= & \\ & -0.5dy \left( \frac{-u[-1+i, -1+j] + u[i, -1+j]}{dx} \right. \\ & \left. + \frac{-u[-1+i, j] + u[i, j]}{dx} \right) + v[-1+j] \end{aligned}$$

O sistema de equações é montado declarando-se as variáveis e montando-se uma tabela das equações na forma discreta.

### ■ Definição da Grade e do Sistema de Equações Discretas

```
In[9]:=
dx = 0.2*N[Degree];      (*passo em x rad*)
dy = 0.1;                (*passo em y*)
limy = 40;               (*N. nós grade y*)
i = 1;
variaveis = Table[u[i,j],{j,1,limy-1}];
eqDiscret = Flatten[Table[eqMovi /.
v[0,j] -> v[j],{j,1,limy-1}]];
```

O Mathematica<sup>®</sup> permite a programação funcional, ou seja, pode-se uma definir funções que, quando alimentadas com um argumento, estas retornam o seu valor. Graças a esta habilidade ficou extremamente simples calcular o gradiente de pressão, bastando definir a função para o perfil de velocidade de Hiemenz, Eq. (5), o qual foi feito da seguinte forma:

### ■ Perfil Externo de Velocidades - Hiemenz & Gradiente de Pressões

```
In[15]:=
Uo[x_] = 1.814*x - 0.271*x^3 - 0.0471*x^5;
gradPress[x_,dx_] = -(1/2)*(Uo[x]^2 -
Uo[x-dx]^2)/dx;
```

As condições de contorno são introduzidas de maneira usual:

### ■ Condições de Contorno

```
In[17]:=
u[1,0] = 0;              (*não deslizamento*)
u[1,limy] := Uo[x];     (*vel. no infinito*)
v[0] = 0;               (*sem sucção/injeção massa*)
```

As condições iniciais foram obtidas solucionando a Eq. (6), utilizando-se de um pacote para solução de equações diferenciais do próprio Mathematica<sup>®</sup>. Também foi realizada a conversão das velocidades, Eqs. (8) e (9) devido a adimensionalização. O procedimento então ficou :

### ■ Condições Iniciais - Solução de Falkner-Skan

Campo de velocidade inicial obtido a partir da solução do problema de estagnação 2-D.

```
In[20]:=
Clear[eqFS,solFS]
eqFS = f'''[h] + f''[h]*f[h] + 1-(f'[h])^2;
solFS = NDSolve[{eqFS==0,f[0]==0,f'[0]==0,
f''[0]==1.23259},f,{h,0,10}][[1,1,2]];
Do[{u[0,j] = 0, v[j] = -(Sqrt[1.814]*
solFS[Sqrt[1.814]*(j)*dy])},{j,0,limy}]
```

O sistema é solucionado calculando as velocidades  $u$  e  $v$  a cada passo  $dx$ . Com o perfil de  $u$  a tensão,  $\tau$ , na parede é determinada por uma diferenciação numérica e armazenada em  $cf$ . Também para cada variação de 1 grau no cilindro é determinada uma função interpolante do perfil de velocidades  $u$ ,  $fe$ , para cálculo da espessura de deslocamento e momento, deltas e teta. Estas grandezas são armazenadas, respectivamente, nas variáveis  $espdes$  e  $espmom$ . Finalmente, os perfis de velocidade  $u$  e  $v$  são gravados em  $Uperfil$  e  $Vperfil$  a cada 10 graus. O procedimento de cálculo avança em  $x$  com passo  $dx$  até o ponto onde ocorre o descolamento, ou seja, quando  $\tau < 0$ .

### ■ Resolução do Sistema

```
In[24]:=
k = 1;
tau = 1;
While[ tau>0 ,
{ x = dx*k;
ang = x/N[Degree];
solução = Solve[eqDiscret,variaveis];
Do[u[1,j] = solução[[1,j,2]],{j,1,limy-1}];
Do[v[j] = eqMassa[j],{j,1,limy}];
Do[u[0,j] = u[1,j],{j,1,limy}];
Do[u[1,j] = .,{j,1,limy-1}];
Clear[sx,fe];
tau = (1/(12*dy))(48*u[0,1] -36*u[0,2] +
16*u[0,3] -3 *u[0,4]);
AppendTo[cf,{ang,tau}];
If[ Chop[Mod[ang,1]] == 0.,
{ fe = Interpolation[Table[ {(j)*dy,
```

```

N[u[0,j]/u[0,limy]] , {j,0,limy}]]];
deltas = NIntegrate[(1-fe[y]),
  {y,0,(limy-1)*dy}];
theta = NIntegrate[fe[y]*(1 -
  fe[y]),{y,0,(limy-1)*dy}];
AppendTo[espdes,{ang,deltas}];
AppendTo[espmom,{ang,theta}];
If[ Chop[Mod[ang,10.]] == 0.,
  {Vperfil[ang] = Table[{(j)*dy,
    v[j]}, {j,0,limy}];
  Uperfil[ang] = Table[{(j)*dy,
    u[0,j]}, {j,0,limy}]]];k++}]]];

```

Após calcular todas as grandezas de interesse, estas são gravadas em arquivos, para posterior análise. Entretanto, antes de gravar é importante apagá-los do disco pois, caso isto não seja feito, haverá um sobreposição de dados aumentando significativamente o tamanho dos arquivos.

### ■ Saída de Dados

```

In[27]:=
DeleteFile["saida1.m"]
DeleteFile["saida2.m"]
DeleteFile["saida3.m"]
Save["saida1.m", espdes, espmom, cf]
Save["saida2.m", Vperfil]
Save["saida3.m", Uperfil]

```

O ponto de descolamento ocorre quando  $cf = 0$ .

### ■ Cálculo do Ponto de Descolamento

```

In[33]:=
Clear[descol,sol]
descol = Interpolation[cf];
sol = FindRoot[descol[dummy] == 0,
  {dummy,{75,ang}}];
Print["Ponto de Descolamento = ",
  N[sol[[1,2]],3], " graus"]

```

Ponto de Descolamento = 80.4 graus

Foi adicionado ao final do programa uma saída gráfica pois, desta forma, fica mais agradável e rápido a visualização dos resultados. Aqui, além de utilizar a habilidade gráfica

do Mathematica<sup>®</sup>, utiliza-se novamente a programação funcional, ou seja, definem-se funções para a geração dos gráficos.

### ■ Saída Gráfica

In[37]:=

```
Clear[graf1,graf2,graf3,majorticks,Cf,theta,
      delta]

CF = Graphics[Text[FontForm["Cf*",
{"Times",14}],{68,1.75}],DisplayFunction -> Identity];
theta = Graphics[Text[FontForm["q",
{"Symbol",18}],{82,0.5}],DisplayFunction -> Identity];
delta = Graphics[Text[FontForm["d",
{"Symbol",18}],{82,1.5}],DisplayFunction -> Identity];

majorticks[min_,max_,step_] :=
Range[Floor[min],Floor[max],step]

graf1[n_] := ListPlot[n,PlotJoined->True,
DisplayFunction->Identity,AxesOrigin->{0,0},
Frame->True,PlotRange->{{0,90},{0,2}},
FrameTicks->{majorticks[0,90,10],majorticks[0,2,0.5]},
GridLines->{majorticks[0,90,10],majorticks[0,2,0.25]},
FrameLabel->{"x (graus)",""}];

graf2[n_] := ListPlot[n,PlotJoined->True,
DisplayFunction->Identity,AxesOrigin->{0,0},
Frame->True,PlotRange->{{0,4},{0,2}},
FrameTicks->{majorticks[0,4,0.5],majorticks[0,2,0.5]},
GridLines->{majorticks[0,4,0.25],majorticks[0,2,0.25]},
FrameLabel->{"y.Re^0.5/R","u/Uinf"}];

graf3[n_] := ListPlot[n,PlotJoined->True,
DisplayFunction->Identity,AxesOrigin->{0,0},
Frame->True,PlotRange->{{0,4},{-3,3}},
FrameTicks->{majorticks[0,4,0.5],majorticks[-3,3,1.0]},
GridLines->{majorticks[0,4,0.25],majorticks[-3,3,0.5]},
FrameLabel->{"y.Re^0.5/R","v.(Re)^0.5/Uinf"}];
```

O coeficiente de atrito  $C_f^*$ , espessura de deslocamento  $\delta^*$  e espessura de momento  $\theta^*$ , em função da posição angular são plotados abaixo, onde  $C_f^*$ ,  $\delta^*$  e  $\theta^*$  estão definidos pelas equações abaixo:


$$C_f^* = \left(\sqrt{\text{Re}_\infty}/R\right) \cdot \tau_w / 0.5 \cdot \rho \cdot U_\infty^2$$

$$\delta^* = \left(\sqrt{\text{Re}_\infty}/R\right) \cdot \int (1-u/U) \cdot dy$$

$$\theta^* = \left( \sqrt{\text{Re}_\infty} / R \right) \cdot \int u/U \cdot (1 - u/U) \cdot dy$$

In[45]:=


```
Show[graf1[cf],CF,graf1[espdes],delta,theta,
graf1[espmom],DefaultFont -> {"Times",14},
DisplayFunction->$DisplayFunction];
```



Os perfis de velocidade  $u/U_\infty$  em função da distância  $y \cdot (\sqrt{\text{Re}_\infty} / R)$  para as posições angulares de  $10^\circ$  em  $10^\circ$  até  $80^\circ$  resultantes são:

In[46]:=


```
Show[graf2[Uperfil[10.]],graf2[Uperfil[20.]],
graf2[Uperfil[30.]],graf2[Uperfil[40.]],
graf2[Uperfil[50.]],graf2[Uperfil[60.]],
graf2[Uperfil[70.]],graf2[Uperfil[80.]],
DefaultFont -> {"Times",14},
DisplayFunction->$DisplayFunction];
```



De forma análoga, os perfis de velocidade  $v \cdot (\sqrt{Re_{\infty}}/R)$  em função da distância  $y \cdot (\sqrt{Re_{\infty}}/R)$  para posições angulares de  $10^{\circ}$  em  $10^{\circ}$  até  $80^{\circ}$  são:

In[47]:=

```
Show[graf3[Vperfil[10.]],graf3[Vperfil[20.]],
graf3[Vperfil[30.]],graf3[Vperfil[40.]],
graf3[Vperfil[50.]],graf3[Vperfil[60.]],
graf3[Vperfil[70.]],graf3[Vperfil[80.]],
DefaultFont -> {"Times",14},
DisplayFunction->$DisplayFunction];
```



## CONCLUSÕES

Este procedimento avalia com precisão os principais parâmetros que caracterizam a camada limite. O ponto de descolamento é previsto em  $80,4^\circ$  enquanto que o valor experimental é de aproximadamente  $80,5^\circ$ . Os valores de  $C_f^*$ ,  $\delta^*$ , e  $\theta^*$  encontram-se em concordância com os valores publicados em Schlichting (1960). Enquanto que os valores dos perfis de velocidades são os mesmos de White (1991). O procedimento permite ainda que, com uma simples mudança nas condições de contorno, possa se estudar a camada limite com sucção ou injeção. escoamentos em outros corpos podem também ser determinados bastando para tanto que se substitua o perfil de velocidades externo por um adequado. Além disto, a simplicidade na obtenção das equações discretizadas facilita a exploração novos esquemas discretizantes.

## Referências & Leitura Complementar

- Blotner, F.G., "Finite Difference Methods of Solution of the Boundary Layer Equations", AIAA J., vol.8, pp. 193-205, 1970
- Cebeci, T., and Bradshaw, P., "Momentum Transfer in Boundary Layers", McGraw Hill, 1977
- Mikhailov, M.D., "Finite Difference Method by Using Mathematica", Int. J. Heat Mass Transf., vol.37, pp.375-379, 1994
- Minkowycz, W.J., Sparrow, E.M., Schneider, G.E. and Pletcher, R.H., "Handbook of Numerical Heat Transfer", John Wiley (1988)
- Patankar, S.V and Spalding, D.B., "Heat and Mass Transfer in Boundary Layers, 2nd ed.", Intertext, London, 1970
- Schlichting, H., "Boundary Layer Theory", McGraw Hill, 1960
- Warsi, Z.U.A., "Fluid Dynamics: Theoretical and Computational Approaches", CRC, 1993
- Wendt, J.F., "Computational Fluid Dynamics: An Introduction", Springer Verlag, 1992
- White, F.M.; "Viscous Fluid Flow, 2nd ed.", McGraw Hill, 1991
- Wolfram, S.; "Mathematica, 2nd ed.", Addison Wesley, 1991